
1 [bookmark: _Toc122434485][bookmark: _Toc110865486]Overview
This document lists all the essencial changes needed to correct problems in the TTCN implementation of NR5GC test case 8.1.6.1.1.2  which is part of the NR5GC test suite in iWD_TTCN3-B2020-09_D22wk23.
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log file(s) is (are) provided as evidence. 
Contact:	Parikshit Bhise
	parikshit.bhise@rohde-schwarz.com 
	
1. [bookmark: _Toc110865487]Verification Test Summary 
Test Case: 	8_1_6_1_1_2_NR5GC
ATS Version:	iwd-TTCN3-B2020-09_D22wk23
[bookmark: _Hlk37919671]System Simulator used:	R&S® 5G Protocol Conformance Test platform
UE used:	MediaTek MT6983
Verification Status:	PASS

2 [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc110865488]Corrections required
2.1 [bookmark: _Toc110865489][bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665]cr_38508_UL_PDCP_DelayValueResult_r16
	Function name
	cr_38508_UL_PDCP_DelayValueResult_r16

	Reason for change
	When measurement results are enabled the UE will always report averageDelay_r16 value =0 when there is not DRB data transmission, therefore it is proposed to parameterise this value and change for non zero value when there is drb data being sent.

	Summary of change
	Added variable p_int to be passed as a parameter in the template

	TTCN module
	RRC_MDT_IntraNR_Immediate_NR5GC

	MCC160 Comment
	Accepted, but Prose CR needed



Before Change:
	template (present) UL_PDCP_DelayValueResult_r16 cr_38508_UL_PDCP_DelayValueResult_r16(DRB_Identity p_DRB_Identity) :=
  { /* @status    */
     drb_Id_r16 := p_DRB_Identity,
     averageDelay_r16 := ?
  };



After Change:
	template (present) UL_PDCP_DelayValueResult_r16 cr_38508_UL_PDCP_DelayValueResult_r16(DRB_Identity p_DRB_Identity,
                                                                                        template (present) integer p_int) := 
  { /* @status    */
     drb_Id_r16 := p_DRB_Identity,
     averageDelay_r16 := p_int 
  };



2.2 [bookmark: _Toc110865490]cs_38508_ReportConfigNR_Periodical_UlDelayValueConfig_r16_setup
	Function name
	cs_38508_ReportConfigNR_Periodical_UlDelayValueConfig_r16_setup

	Reason for change
	The default drb established is DRB2 and the 2 additional DRBs that are established will be 3 and 4, therefore the delay_DRBlist needs to be updated with correct 3 and 4 where the loop back data is sent out.

	Summary of change
	Modified the delayed delay_DRBlist to {3,4}

	TTCN module
	RRC_MDT_IntraNR_Immediate_NR5GC

	MCC160 Comment
	Accepted, but Prose CR needed



Before Change:
	template (value) ReportConfigNR cs_38508_ReportConfigNR_Periodical_UlDelayValueConfig_r16_setup
  :=
  { /* @status    */
    reportType := {
      periodical := {
        rsType := ssb,
        reportInterval := ms480,    //Acc Table 4.6.3-109 TS 38.508-1
        reportAmount := infinity_,
        reportQuantityCell := cs_NR_MeasReportQuantity_All,
        maxReportCells := 8,
        reportQuantityRS_Indexes := omit,
        maxNrofRS_IndexesToReport := omit,
        includeBeamMeasurements := false,
        useWhiteCellList := false,
        measRSSI_ReportConfig_r16     := omit,   /* @sic R5s201387 BASELINE MOVING 2020 sic@ */
        includeCommonLocationInfo_r16 := omit,   /* @sic R5s201387 BASELINE MOVING 2020 sic@ */
        includeBT_Meas_r16            := omit,   /* @sic R5s201387 BASELINE MOVING 2020 sic@ */
        includeWLAN_Meas_r16          := omit,   /* @sic R5s201387 BASELINE MOVING 2020 sic@ */
        includeSensor_Meas_r16        := omit,   /* @sic R5s201387 BASELINE MOVING 2020 sic@ */
        ul_DelayValueConfig_r16       :=
        {
           setup :=
           {
              delay_DRBlist := {2, 3}
           }
        },
        reportAddNeighMeas_r16        := omit    /* @sic R5s201387 BASELINE MOVING 2020 sic@ */
      }
    }
  };



After Change:
	  template (value) ReportConfigNR cs_38508_ReportConfigNR_Periodical_UlDelayValueConfig_r16_setup
  :=
  { /* @status    */
    reportType := {
      periodical := {
        rsType := ssb,
        reportInterval := ms480,    //Acc Table 4.6.3-109 TS 38.508-1
        reportAmount := infinity_,
        reportQuantityCell := cs_NR_MeasReportQuantity_All,
        maxReportCells := 8,
        reportQuantityRS_Indexes := omit,
        maxNrofRS_IndexesToReport := omit,
        includeBeamMeasurements := false,
        useWhiteCellList := false,
        measRSSI_ReportConfig_r16     := omit,   /* @sic R5s201387 BASELINE MOVING 2020 sic@ */
        includeCommonLocationInfo_r16 := omit,   /* @sic R5s201387 BASELINE MOVING 2020 sic@ */
        includeBT_Meas_r16            := omit,   /* @sic R5s201387 BASELINE MOVING 2020 sic@ */
        includeWLAN_Meas_r16          := omit,   /* @sic R5s201387 BASELINE MOVING 2020 sic@ */
        includeSensor_Meas_r16        := omit,   /* @sic R5s201387 BASELINE MOVING 2020 sic@ */
        ul_DelayValueConfig_r16       :=
        {
           setup :=
           {
              delay_DRBlist :=  {3,4} //
           }
        },
        reportAddNeighMeas_r16        := omit    /* @sic R5s201387 BASELINE MOVING 2020 sic@ */
      }
    }
  };



2.3 [bookmark: _Toc110865491]f_TC_8_1_6_1_1_2_NR5GC
	Function name
	f_TC_8_1_6_1_1_2_NR5GC

	Reason for change
	1. The QFI values configured in the v_SS_Drb_ConfigList does not match the QFIs configured in the PDU Session establishment accept message

2. f_NR5GC_RRC_ConnectedState3N_Def is called twice in the code (duplicate) and as part of the test IP data is exchanged and therefore the loop back needs to be configured as TEST_LOOPModeB

3. The postamble call for this test case is missing at the end of the test cases


	Summary of change
	1. Modified the QFIs to use DRB 2 (first DRB) – QFI 1, DRB 3 (second DRB) – QF5 1, DRB 4 (Third DRB) – QFI 2
2. Removed the duplicate call for f_NR5GC_RRC_ConnectedState3N_Def and modified the code to use  TEST_LOOPModeB and added v_IP_PDU_delay
3. Added f_NR_Postamble(nr_Cell1, STATE_IDLE_1A);

	TTCN module
	RRC_MDT_IntraNR_Immediate_NR5GC

	MCC160 Comment
	Change 1: Accepted
Change 2: Accepted
Change 3: Accepted



Before Change:
	function f_TC_8_1_6_1_1_2_NR5GC() runs on NR5GC_PTC
  { //Immediate MDT / Measurement / Latency metrics for UL PDCP Packet Delay per DRB
   
    var Octet_Type v_IP_PDU_delay := '05'O;
    var IntegerList_Type v_DRBList;
    var integer v_FirstDRB;
    var integer v_SecondDRB;
    var integer v_ThirdDRB;
    var integer v_PduSessionId;
    var GSM_MobilityInfo_Type v_GSM_MobilityInfo_Type;
    var template (value) NR_RadioBearerList_Type v_SS_Drb_ConfigList;
    var template (value) DRB_ToAddModList v_DRB_ToAddModList;
    var template (value) QoS_Rules v_QoS_Rules;
    var template (value) QoSFlowDescr v_QoSFlowDescr;
    var template (value) QosFlow_Identification_Type v_QosFlowId;
    var PDN_Index_Type v_PDN_Index;
    var template (value) QoS_Rule v_QosRule1;
    var template (value) QoS_Rule v_QosRule4;
    
    f_NR5GC_Init(NR_1);
    
    //Create and configure cell
    f_NR_CellConfig_Def(nr_Cell1);
    
    // The UE is in 5GS state 3N-A with one PDU session active according to TS 38.508-1 [4], clause 4.4A.3 Table 4.4A.3-1.
    // and using the message condition UE TEST LOOP MODE A active. 3 DRBs are configured where DRB 1 is defined as default DRB.
    // Perform Registration and get PDU Establishment Request message
    v_GSM_MobilityInfo_Type := f_MultipleQoSPreamble_Part1 (nr_Cell1);
    v_PduSessionId := oct2int(v_GSM_MobilityInfo_Type.SessionId);
    
    //Get DRB identities
    v_DRBList:= f_NR_GetNextDRBIds(v_GSM_MobilityInfo_Type, true, 3); // First is Default DRB
    v_FirstDRB := v_DRBList[0];
    v_SecondDRB := v_DRBList[1];
    v_ThirdDRB := v_DRBList[2];

    //Note: In order to be able to start the IP handling during the preamble, QFI=1 is mapped temporarily to the default DRB (v_FirtDRB).
    //      This mapping will be removed when moving to connected mode
    v_SS_Drb_ConfigList := {f_NR_GetRadioBearerConfig_FullAM(nr_Cell1,
                                                             v_FirstDRB,
                                                             -, //RLC normal mode
                                                             -, //PDCP normal mode
                                                             -,
                                                             cs_SDAP_Configuration(v_PduSessionId, -, -, {1,2})), //Map QFI=1 to default in DRB to deal with initial IMS registration
                            f_NR_GetRadioBearerConfig_FullAM(nr_Cell1,
                                                             v_SecondDRB,
                                                             -, //RLC normal mode
                                                             -, //PDCP normal mode
                                                             -,
                                                             cs_SDAP_Configuration(v_PduSessionId, -, -, {5})),
                            f_NR_GetRadioBearerConfig_FullAM(nr_Cell1,
                                                             v_ThirdDRB,
                                                             -, //RLC normal mode
                                                             -, //PDCP normal mode
                                                             -,
                                                             cs_SDAP_Configuration(v_PduSessionId, -, -, {8}))};
                                                         
    v_DRB_ToAddModList := {cs_38508_NRDRB_ToAddMod(cs_SDAP_Config(v_PduSessionId, true, {2}),
                                                   v_FirstDRB,
                                                   cs_38508_PDCP_Config(ms300)),
                           cs_38508_NRDRB_ToAddMod(cs_SDAP_Config(v_PduSessionId, false, {5}),
                                                   v_SecondDRB,
                                                   cs_38508_PDCP_Config(ms300)),
                           cs_38508_NRDRB_ToAddMod(cs_SDAP_Config(v_PduSessionId, false, {8}),
                                                   v_ThirdDRB,
                                                   cs_38508_PDCP_Config(ms300))};
                                              
    v_QosRule1 := f_BuildDefaultQoSRule ("1");
    v_QosRule4 := f_BuildQoSRuleRemoteAccess("4");
    v_QosRule4.dqrBit := '0'B;
    
    v_QoS_Rules := cs_QoS_Rules ({v_QosRule1, v_QosRule4, f_BuildQoSRuleRemoteAccess("5")});
    v_QoSFlowDescr := cs_QoSFlowDescr ({cs_QoS_Flow1, cs_QoS_Flow2, cs_QoS_Flow3});

    // Start IP handling in order for the TTCN to handle IPv6 address allocation,
    // then stop it before the test body, so that TTCN can route the IP data
    v_PDN_Index := f_NR5GC_GetPdnIndex(v_GSM_MobilityInfo_Type);
    v_QosFlowId := cs_IP_QosFlowId(oct2int(v_GSM_MobilityInfo_Type.SessionId), 1); //Picked one QoS flow associated with the default DRB to start the IP handling
    f_IP_Handling_StartPDN(IP, v_PDN_Index, cs_DrbInfo_NR(nr_Cell1, v_QosFlowId));
  
    f_MultipleQoSPreamble_Part3 (nr_Cell1,
                                 v_GSM_MobilityInfo_Type,
                                 v_FirstDRB,
                                 v_SecondDRB,
                                 v_ThirdDRB,
                                 v_DRB_ToAddModList,
                                 v_SS_Drb_ConfigList,
                                 v_QoS_Rules,
                                 v_QoSFlowDescr,
                                 v_PDN_Index,
                                 false);
                            
    f_NR_AS_CipheringAlgorithm_Set(nea0);
   
    f_NR5GC_RRC_ConnectedState3N_Def(nr_Cell1,
                                     TEST_LOOPModeA_ON,
                                     -,-,-,-,
                                     v_DRB_ToAddModList,
                                     v_SS_Drb_ConfigList);
    
    f_NR5GC_RRC_ConnectedState3N_Def(nr_Cell1, TEST_LOOPModeA_ON, -, v_IP_PDU_delay);

    f_NR_TestBody_Set(true);

    f_TC_8_1_6_1_1_2_NR5GC_TestBody(v_PduSessionId);

    f_NR_TestBody_Set(false);

  } //end of f_TC_8_1_6_1_1_2_NR5GC



After Change:
	function f_TC_8_1_6_1_1_2_NR5GC() runs on NR5GC_PTC
  { //Immediate MDT / Measurement / Latency metrics for UL PDCP Packet Delay per DRB
   
    var Octet_Type v_IP_PDU_delay := '05'O;
    var IntegerList_Type v_DRBList;
    var integer v_FirstDRB;
    var integer v_SecondDRB;
    var integer v_ThirdDRB;
    var integer v_PduSessionId;
    var GSM_MobilityInfo_Type v_GSM_MobilityInfo_Type;
    var template (value) NR_RadioBearerList_Type v_SS_Drb_ConfigList;
    var template (value) DRB_ToAddModList v_DRB_ToAddModList;
    var template (value) QoS_Rules v_QoS_Rules;
    var template (value) QoSFlowDescr v_QoSFlowDescr;
    var template (value) QosFlow_Identification_Type v_QosFlowId;
    var PDN_Index_Type v_PDN_Index;
    var template (value) QoS_Rule v_QosRule1;
    var template (value) QoS_Rule v_QosRule4;
    
    f_NR5GC_Init(NR_1);
    
    //Create and configure cell
    f_NR_CellConfig_Def(nr_Cell1);
    
    // The UE is in 5GS state 3N-A with one PDU session active according to TS 38.508-1 [4], clause 4.4A.3 Table 4.4A.3-1.
    // and using the message condition UE TEST LOOP MODE A active. 3 DRBs are configured where DRB 1 is defined as default DRB.
    // Perform Registration and get PDU Establishment Request message
    v_GSM_MobilityInfo_Type := f_MultipleQoSPreamble_Part1 (nr_Cell1);
    v_PduSessionId := oct2int(v_GSM_MobilityInfo_Type.SessionId);
    
    //Get DRB identities
    v_DRBList:= f_NR_GetNextDRBIds(v_GSM_MobilityInfo_Type, true, 3); // First is Default DRB
    v_FirstDRB := v_DRBList[0];
    v_SecondDRB := v_DRBList[1];
    v_ThirdDRB := v_DRBList[2];

    //Note: In order to be able to start the IP handling during the preamble, QFI=1 is mapped temporarily to the default DRB (v_FirtDRB).
    //      This mapping will be removed when moving to connected mode
    v_SS_Drb_ConfigList := {f_NR_GetRadioBearerConfig_FullAM(nr_Cell1,
                                                             v_FirstDRB,
                                                             -, //RLC normal mode
                                                             -, //PDCP normal mode
                                                             -,
                                                             cs_SDAP_Configuration(v_PduSessionId, -, -, {1})), //Map QFI=1 to default in DRB to deal with initial IMS registration 
                            f_NR_GetRadioBearerConfig_FullAM(nr_Cell1,
                                                             v_SecondDRB,
                                                             -, //RLC normal mode
                                                             -, //PDCP normal mode
                                                             -,
                                                             cs_SDAP_Configuration(v_PduSessionId, -, -, {5})),
                            f_NR_GetRadioBearerConfig_FullAM(nr_Cell1,
                                                             v_ThirdDRB,
                                                             -, //RLC normal mode
                                                             -, //PDCP normal mode
                                                             -,
                                                             cs_SDAP_Configuration(v_PduSessionId, -, -, {2}))}; 
                                                         
    v_DRB_ToAddModList := {cs_38508_NRDRB_ToAddMod(cs_SDAP_Config(v_PduSessionId, true, {1}), //WA#WI=866597 changed to use 1
                                                   v_FirstDRB,
                                                   cs_38508_PDCP_Config(ms300)),
                           cs_38508_NRDRB_ToAddMod(cs_SDAP_Config(v_PduSessionId, false, {5}), 
                                                   v_SecondDRB,
                                                   cs_38508_PDCP_Config(ms300)),
                           cs_38508_NRDRB_ToAddMod(cs_SDAP_Config(v_PduSessionId, false, {2}), // changed to use 2
                                                   v_ThirdDRB,
                                                   cs_38508_PDCP_Config(ms300))};
                                              
    v_QosRule1 := f_BuildDefaultQoSRule ("1");
    v_QosRule4 := f_BuildQoSRuleRemoteAccess("4");
    v_QosRule4.dqrBit := '0'B;
    
    v_QoS_Rules := cs_QoS_Rules ({v_QosRule1, v_QosRule4, f_BuildQoSRuleRemoteAccess("5")});
    v_QoSFlowDescr := cs_QoSFlowDescr ({cs_QoS_Flow1, cs_QoS_Flow2, cs_QoS_Flow3});

    // Start IP handling in order for the TTCN to handle IPv6 address allocation,
    // then stop it before the test body, so that TTCN can route the IP data
    v_PDN_Index := f_NR5GC_GetPdnIndex(v_GSM_MobilityInfo_Type);
    v_QosFlowId := cs_IP_QosFlowId(oct2int(v_GSM_MobilityInfo_Type.SessionId), 1); //Picked one QoS flow associated with the default DRB to start the IP handling
    f_IP_Handling_StartPDN(IP, v_PDN_Index, cs_DrbInfo_NR(nr_Cell1, v_QosFlowId));
  
    f_MultipleQoSPreamble_Part3 (nr_Cell1,
                                 v_GSM_MobilityInfo_Type,
                                 v_FirstDRB,
                                 v_SecondDRB,
                                 v_ThirdDRB,
                                 v_DRB_ToAddModList,
                                 v_SS_Drb_ConfigList,
                                 v_QoS_Rules,
                                 v_QoSFlowDescr,
                                 v_PDN_Index,
                                 false);
                            
    f_NR_AS_CipheringAlgorithm_Set(nea0);
     //WA#WI=866597
    f_NR5GC_RRC_ConnectedState3N_Def(nr_Cell1,
                                     TEST_LOOPModeB_ON, //WA#WI=866597 changed to looped back mode B as IP packets are sent/ received
                                    -,v_IP_PDU_delay,-,-,
                                     v_DRB_ToAddModList,
                                     v_SS_Drb_ConfigList);
    
    //f_NR5GC_RRC_ConnectedState3N_Def(nr_Cell1, TEST_LOOPModeA_ON, -, v_IP_PDU_delay); duplicate call

    f_NR_TestBody_Set(true);

    f_TC_8_1_6_1_1_2_NR5GC_TestBody(v_PduSessionId);

    f_NR_TestBody_Set(false);
    f_NR_Postamble(nr_Cell1, STATE_IDLE_1A); //WA#WI=866597

  } //end of f_TC_8_1_6_1_1_2_NR5GC



2.4 [bookmark: _Toc110865492]f_TC_8_1_6_1_1_2_NR5GC_TestBody
	Function name
	f_TC_8_1_6_1_1_2_NR5GC_TestBody

	Reason for change
	1. According to the changes in 2.3 aligned the test case to use QFI 2 (instead of QFI8) and the corresponding IP data variable / and QoS id updated
2. The measurement configuration configures periodic measurements for reportQuantityCell (rsrp, rsrq..) and this is periodically reported through out the test case and needs to be handled
3. When measurement configuration is configured with MeasConfig to setup ul-DelayValueConfig, the UE shall start to report a value for averageDelay_r16, therefore it is prposed to check at Step 5A that the PDCP-DelayValueResult is having a non zero value for the DRB where the data is loopedback (similarly for Step 8) 
4. At Step 9  the RRCReconfiguration message releases the DelayValueConfig measurement, but the periodic measurements for reportQuantityCell (rsrp, rsrq..) is still active and therefore the UE shall send periodict measurement report with measresults with resultsSSB_Cell and no ul_PDCP_DelayValueResultList_r16
5. The measurement configuration configures periodic measurements for reportQuantityCell (rsrp, rsrq..) and this is periodically reported and in order to avoid handling the measurement report message during postamble, it is proposed to release the RRC connection before entering the postamble

	Summary of change
	1. Renamed the variable to v_IPDataQFI2 and v_QoS_Id2 to be used through out the test case, v_IPDataQFI2 is created based on match filter 3
2. Created a new default v_DefaultRef and called the function activate(a_NR_AddDefault_IgnorePeriodicalMeasurmentReport(nr_Cell1,cr_NR_MeasurementReport_Any)); to handle the periodic measurement report that is received through out the test cases. In cases where we are specifically checking for the measurement report, this default is deactivated and specification function are used to check the measurement reports
3. v_MeasResults is modified to check a non zero value for DRB 3 and 0 for DRB 4 at Step 5A and a new function is called to check the reported measurement values new function f_NR_ReceivePeriodicMeasurementReport (see change 2.6). At Step 8 v_MeasResults is modified to check for non zero value for DRB 4 and 0 for DRB 3 and new function f_NR_ReceivePeriodicMeasurementReport is used to check the measurement report.
4. Added new function f_NR_Check_NoDelayValueMeasurementReport (See change 2.5) and provided variables v_MeasResults, v_MeasResults_NoDelayValue to check for themeasurement report and assign fail verdict accordingly.
5. Added f_NR_RRCRelease(nr_Cell1);

	TTCN module
	RRC_MDT_IntraNR_Immediate_NR5GC

	MCC160 Comment
	Change 1: Accepted, but Prose CR needed
Change 2: Accepted, but Prose CR needed
Change 3: Accepted, but Prose CR needed
Change 4: Accepted, but Prose CR needed
Change 5: Accepted



Before Change:
	function f_TC_8_1_6_1_1_2_NR5GC_TestBody(integer p_PduSessionId) runs on NR_BASE_PTC
  {

   var PhysCellId v_PhysCellId;
   var RSRP_Range v_AbsThreshSS_Rsrp;
      
   var template (value) MeasObjectParamsList_Type v_MeasuredObjectList;
   var template (value) ReportConfigToAddModList v_ReportConfigList;
   var template (value) MeasIdToAddModList v_MeasIdConfigList;
   var template (omit) MeasConfig v_MeasConfig;
   var NR_UplinkBWP_Type v_NR_UplinkBWP;
   var NR_ResourceAllocation_Type v_NR_ResourceAllocation;
   var template (present) MeasResults v_MeasResults;
   
   var octetstring v_IPDataQFI5 := f_SdapTests_CreateDatagram("4"); // Packet with remote address matching filter #4
   var octetstring v_IPDataQFI8 := f_SdapTests_CreateDatagram("5"); // Packet with remote address matching filter #5
   var template (value) QosFlow_Identification_Type v_QoS_Id5 := cs_IP_QosFlowId(p_PduSessionId, 5); // QFI5 is mapped to Second DRB
   var template (value) QosFlow_Identification_Type v_QoS_Id8 := cs_IP_QosFlowId(p_PduSessionId, 8); // QFI8 is mapped to Third DRB

   //@siclog "Steps 1-2" siclog@
   // The SS transmits an RRCReconfiguration message including MeasConfig to setup ul-DelayValueConfig
   // The UE transmits an RRCReconfigurationComplete message.
   v_PhysCellId := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);
   v_AbsThreshSS_Rsrp := 57;
    
   v_MeasuredObjectList := {cs_MeasObjectId1(nr_Cell1, v_AbsThreshSS_Rsrp)};
   v_ReportConfigList := {cs_NR_ReportConfig_Periodical_UlDelayValueConfig_r16_setup(tsc_NR_IdReportConfigId1)};
   v_MeasIdConfigList := {cs_NR_MeasId_Config_id1_obj1_conf1};
      
   v_MeasConfig :=f_NR_GenerateMeasurementConfigNR(v_MeasuredObjectList, v_ReportConfigList, v_MeasIdConfigList);
      
   f_NR_SendRRCReconfiguration_MeasNR(nr_Cell1, v_MeasConfig);
   f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");
   
   //@siclog "Step 3" siclog@
   // The SS transmits UL grant to the UE at every 5ms in PDCCH occasion
   v_NR_UplinkBWP := f_NR_CellInfo_GetUplinkBWP(nr_Cell1, tsc_NR_BWP_Id);
   v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(136, v_NR_UplinkBWP);
   f_NR_ULGrantConfiguration_StartPeriodicGrant(nr_Cell1, cs_TimingInfo_Now, v_NR_ResourceAllocation, 5); //@sic R5-223264 sic@
   
   //@siclog "Step 4" siclog@
   // The SS transmits an IP Packet on the data radio bearer associated with the dedicated 5GC bearer contex #2
   DRB.send(cas_NR_DRB_COMMON_REQ_SDAP_SDUList(nr_Cell1, v_QoS_Id5, {v_IPDataQFI5}));
    
   //@siclog "Step 5" siclog@
   // The UE loop backs the IP packet received in step 4 on DRB 2
   DRB.receive(car_NR_DRB_COMMON_IND_SDAP_SDUList(nr_Cell1, v_QoS_Id5, {v_IPDataQFI5}));
   
   //@siclog "Step 5A" siclog@
   // Check: Does the UE transmit a MeasurementReport message containing UL-PDCP-DelayValueResult for DRB#2?
   v_MeasResults := cr_38508_MeasResults_Ul_PDCP_DelayValueResultList(-, v_PhysCellId, -, -, {cr_38508_UL_PDCP_DelayValueResult_r16(2)}); //@sic R5-223264 sic@
   f_NR_ReceiveOneMeasurementReport(nr_Cell1, v_MeasResults);
   f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 5A");
   
   //@siclog "Step 6" siclog@
   // The SS transmits an IP Packet on the data radio bearer associated with the dedicated 5GC bearer contex #3
   DRB.send(cas_NR_DRB_COMMON_REQ_SDAP_SDUList(nr_Cell1, v_QoS_Id8, {v_IPDataQFI8}));
       
   //@siclog "Step 7" siclog@
   // The UE loop backs the IP packet received in step 6 on DRB 3
   DRB.receive(car_NR_DRB_COMMON_IND_SDAP_SDUList(nr_Cell1, v_QoS_Id8, {v_IPDataQFI8}));
   
   //@siclog "Step 8" siclog@
   // Check: Does the UE transmit a MeasurementReport message containing UL-PDCP-DelayValueResult for DRB#2 and DRB#3
   v_MeasResults := cr_38508_MeasResults_Ul_PDCP_DelayValueResultList(-, v_PhysCellId, -, -, {cr_38508_UL_PDCP_DelayValueResult_r16(2), cr_38508_UL_PDCP_DelayValueResult_r16(3)});
   f_NR_ReceiveOneMeasurementReport(nr_Cell1, v_MeasResults);
   f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 8");
   
   //@siclog "Steps 9-10" siclog@
   // The SS transmits an RRCReconfiguration message including MeasConfig to release the ul-DelayValueConfig measurement
   // The UE transmits an RRCReconfigurationComplete message.
   v_MeasuredObjectList := {cs_MeasObjectId1(nr_Cell1, v_AbsThreshSS_Rsrp)};
   v_ReportConfigList := {cs_NR_ReportConfig_Periodical_UlDelayValueConfig_r16_release(tsc_NR_IdReportConfigId1)};
   v_MeasIdConfigList := {cs_NR_MeasId_Config_id1_obj1_conf1};
      
   v_MeasConfig :=f_NR_GenerateMeasurementConfigNR(v_MeasuredObjectList, v_ReportConfigList, v_MeasIdConfigList);
      
   f_NR_SendRRCReconfiguration_MeasNR(nr_Cell1, v_MeasConfig);
   
   //@siclog "Steps  11-14 " siclog@
   //Parallel behaviour
   //Check: Does the UE transmit a MeasurementReport message containing UL-PDCP-DelayValueResult?
   v_MeasResults := cr_38508_MeasResults_Ul_PDCP_DelayValueResultList(-, v_PhysCellId, -, -, {cr_38508_UL_PDCP_DelayValueResult_r16(2)}); //@sic R5-222264 sic@
   f_NR_Check_NoMeasurementReport(10.0, nr_Cell1, cr_NR_MeasurementReport(v_MeasResults)); //@sic R5-222264 sic@
   
   //@siclog "Step 11" siclog@
   // The SS transmits an IP Packet on the data radio bearer associated with the dedicated 5GC bearer contex #2
   DRB.send(cas_NR_DRB_COMMON_REQ_SDAP_SDUList(nr_Cell1, v_QoS_Id5, {v_IPDataQFI5}));
   
   //@siclog "Step 12" siclog@
   // The UE loop backs the IP packet received in step 4 on DRB 2
   DRB.receive(car_NR_DRB_COMMON_IND_SDAP_SDUList(nr_Cell1, v_QoS_Id5, {v_IPDataQFI5}));
   
   //@siclog "Step 13" siclog@
   // The SS transmits an IP Packet on the data radio bearer associated with the dedicated 5GC bearer contex #3
   DRB.send(cas_NR_DRB_COMMON_REQ_SDAP_SDUList(nr_Cell1, v_QoS_Id8, {v_IPDataQFI8}));
      
   //@siclog "Step 14" siclog@
   // The UE loop backs the IP packet received in step 13 on DRB 3
   DRB.receive(car_NR_DRB_COMMON_IND_SDAP_SDUList(nr_Cell1, v_QoS_Id8, {v_IPDataQFI8}));

 }//end of f_TC_8_1_6_1_1_2_NR5GC_TestBody



After Change:
	function f_TC_8_1_6_1_1_2_NR5GC_TestBody(integer p_PduSessionId) runs on NR_BASE_PTC
  {

   var PhysCellId v_PhysCellId;
   var RSRP_Range v_AbsThreshSS_Rsrp;
      
   var template (value) MeasObjectParamsList_Type v_MeasuredObjectList;
   var template (value) ReportConfigToAddModList v_ReportConfigList;
   var template (value) MeasIdToAddModList v_MeasIdConfigList;
   var template (omit) MeasConfig v_MeasConfig;
   var NR_UplinkBWP_Type v_NR_UplinkBWP;
   var NR_ResourceAllocation_Type v_NR_ResourceAllocation;
   var template (present) MeasResults v_MeasResults;
   var template (present) MeasResults v_MeasResults_NoDelayValue; //WA#WI=866597
   var octetstring v_IPDataQFI5 := f_SdapTests_CreateDatagram("4"); // Packet with remote address matching filter #4
   var octetstring v_IPDataQFI2 := f_SdapTests_CreateDatagram("3"); // Packet with remote address matching filter #3 //WA#WI=866597
   var template (value) QosFlow_Identification_Type v_QoS_Id5 := cs_IP_QosFlowId(p_PduSessionId, 5); // QFI5 is mapped to Second DRB
   var template (value) QosFlow_Identification_Type v_QoS_Id2 := cs_IP_QosFlowId(p_PduSessionId, 2); // QFI2 is mapped to Third DRB //WA#WI=866597
   var default v_DefaultRef;  //WA#WI=866597
   //@siclog "Steps 1-2" siclog@
   // The SS transmits an RRCReconfiguration message including MeasConfig to setup ul-DelayValueConfig
   // The UE transmits an RRCReconfigurationComplete message.
   v_PhysCellId := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);
   v_AbsThreshSS_Rsrp := 57;
    
   v_MeasuredObjectList := {cs_MeasObjectId1(nr_Cell1, v_AbsThreshSS_Rsrp)};
   v_ReportConfigList := {cs_NR_ReportConfig_Periodical_UlDelayValueConfig_r16_setup(tsc_NR_IdReportConfigId1)};
   v_MeasIdConfigList := {cs_NR_MeasId_Config_id1_obj1_conf1};
   v_DefaultRef := activate(a_NR_AddDefault_IgnorePeriodicalMeasurmentReport(nr_Cell1,cr_NR_MeasurementReport_Any)); //WA#WI=866597  
   v_MeasConfig :=f_NR_GenerateMeasurementConfigNR(v_MeasuredObjectList, v_ReportConfigList, v_MeasIdConfigList);
      
   f_NR_SendRRCReconfiguration_MeasNR(nr_Cell1, v_MeasConfig);
   f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");
   
   //@siclog "Step 3" siclog@
   // The SS transmits UL grant to the UE at every 5ms in PDCCH occasion
   v_NR_UplinkBWP := f_NR_CellInfo_GetUplinkBWP(nr_Cell1, tsc_NR_BWP_Id);
   v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(136, v_NR_UplinkBWP);
   f_NR_ULGrantConfiguration_StartPeriodicGrant(nr_Cell1, cs_TimingInfo_Now, v_NR_ResourceAllocation, 5); //@sic R5-223264 sic@
   
   //@siclog "Step 4" siclog@
   // The SS transmits an IP Packet on the data radio bearer associated with the dedicated 5GC bearer contex #2
   DRB.send(cas_NR_DRB_COMMON_REQ_SDAP_SDUList(nr_Cell1, v_QoS_Id5, {v_IPDataQFI5}));
    
   //@siclog "Step 5" siclog@
   // The UE loop backs the IP packet received in step 4 on DRB 2
   DRB.receive(car_NR_DRB_COMMON_IND_SDAP_SDUList(nr_Cell1, v_QoS_Id5, {v_IPDataQFI5}));
   
   //@siclog "Step 5A" siclog@
   // Check: Does the UE transmit a MeasurementReport message containing UL-PDCP-DelayValueResult for DRB#2? //WA#WI=866597
   v_MeasResults := cr_38508_MeasResults_Ul_PDCP_DelayValueResultList(-, v_PhysCellId, -, -, {cr_38508_UL_PDCP_DelayValueResult_r16(3,(1..10000)),cr_38508_UL_PDCP_DelayValueResult_r16(4,0)}); //@sic R5-223264 sic@
   deactivate(v_DefaultRef);   //WA#WI=866597 
   
   f_NR_ReceivePeriodicMeasurementReport(nr_Cell1, v_MeasResults); //WA#WI=866597 to receive periodic measurement report and match one of them
   f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 5A");
   v_DefaultRef := activate(a_NR_AddDefault_IgnorePeriodicalMeasurmentReport(nr_Cell1,cr_NR_MeasurementReport_Any)); //WA#WI=866597 
   //@siclog "Step 6" siclog@
   // The SS transmits an IP Packet on the data radio bearer associated with the dedicated 5GC bearer contex #3
   DRB.send(cas_NR_DRB_COMMON_REQ_SDAP_SDUList(nr_Cell1, v_QoS_Id2, {v_IPDataQFI2})); //WA#WI=866597
       
   //@siclog "Step 7" siclog@
   // The UE loop backs the IP packet received in step 6 on DRB 3
   DRB.receive(car_NR_DRB_COMMON_IND_SDAP_SDUList(nr_Cell1, v_QoS_Id2, {v_IPDataQFI2})); //WA#WI=866597
   
   //@siclog "Step 8" siclog@
   // Check: Does the UE transmit a MeasurementReport message containing UL-PDCP-DelayValueResult for DRB#2 and DRB#3
   v_MeasResults := cr_38508_MeasResults_Ul_PDCP_DelayValueResultList(-, v_PhysCellId, -, -, {cr_38508_UL_PDCP_DelayValueResult_r16(3,0), cr_38508_UL_PDCP_DelayValueResult_r16(4, (1..10000))}); //WA#WI=866597 drb id updated
   deactivate(v_DefaultRef);   //WA#WI=866597

   f_NR_ReceivePeriodicMeasurementReport(nr_Cell1, v_MeasResults);
   f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 8");
   v_DefaultRef := activate(a_NR_AddDefault_IgnorePeriodicalMeasurmentReport(nr_Cell1,cr_NR_MeasurementReport_Any)); //WA#WI=866597
   //@siclog "Steps 9-10" siclog@
   // The SS transmits an RRCReconfiguration message including MeasConfig to release the ul-DelayValueConfig measurement
   // The UE transmits an RRCReconfigurationComplete message.
   v_MeasuredObjectList := {cs_MeasObjectId1(nr_Cell1, v_AbsThreshSS_Rsrp)};
   v_ReportConfigList := {cs_NR_ReportConfig_Periodical_UlDelayValueConfig_r16_release(tsc_NR_IdReportConfigId1)};
   v_MeasIdConfigList := {cs_NR_MeasId_Config_id1_obj1_conf1};
      
   v_MeasConfig :=f_NR_GenerateMeasurementConfigNR(v_MeasuredObjectList, v_ReportConfigList, v_MeasIdConfigList);
      
   f_NR_SendRRCReconfiguration_MeasNR(nr_Cell1, v_MeasConfig);
   
   //@siclog "Steps  11-14 " siclog@
   //Parallel behaviour
   //Check: Does the UE transmit a MeasurementReport message containing UL-PDCP-DelayValueResult?
   //WA#WI=866597
   v_MeasResults := cr_38508_MeasResults_Ul_PDCP_DelayValueResultList(-, v_PhysCellId, -, -, ?); //@sic R5-222264 sic@
   v_MeasResults_NoDelayValue := cr_38508_MeasResults_Ul_PDCP_DelayValueResultList(-, v_PhysCellId, -, -, omit); //@sic R5-222264 sic@
   deactivate(v_DefaultRef);   //WA#WI=866597
   f_NR_Check_NoDelayValueMeasurementReport(10.0, nr_Cell1, v_MeasResults,v_MeasResults_NoDelayValue ); //@sic R5-222264 sic@
   v_DefaultRef := activate(a_NR_AddDefault_IgnorePeriodicalMeasurmentReport(nr_Cell1,cr_NR_MeasurementReport_Any)); //WA#WI=866597 
   //@siclog "Step 11" siclog@
   // The SS transmits an IP Packet on the data radio bearer associated with the dedicated 5GC bearer contex #2
   DRB.send(cas_NR_DRB_COMMON_REQ_SDAP_SDUList(nr_Cell1, v_QoS_Id5, {v_IPDataQFI5}));
   
   //@siclog "Step 12" siclog@
   // The UE loop backs the IP packet received in step 4 on DRB 2
   DRB.receive(car_NR_DRB_COMMON_IND_SDAP_SDUList(nr_Cell1, v_QoS_Id5, {v_IPDataQFI5}));
   
   //@siclog "Step 13" siclog@
   // The SS transmits an IP Packet on the data radio bearer associated with the dedicated 5GC bearer contex #3
   DRB.send(cas_NR_DRB_COMMON_REQ_SDAP_SDUList(nr_Cell1, v_QoS_Id2, {v_IPDataQFI2})); //WA#WI=866597
      
   //@siclog "Step 14" siclog@
   // The UE loop backs the IP packet received in step 13 on DRB 3
   DRB.receive(car_NR_DRB_COMMON_IND_SDAP_SDUList(nr_Cell1, v_QoS_Id2, {v_IPDataQFI2})); //WA#WI=866597
   f_NR_RRCRelease(nr_Cell1); //WA#WI=866597
   deactivate(v_DefaultRef);   //WA#WI=866597
 }//end of f_TC_8_1_6_1_1_2_NR5GC_TestBody
    

	



2.5 [bookmark: _Toc110865493]f_NR_Check_NoDelayValueMeasurementReport 
	Function name
	f_NR_Check_NoDelayValueMeasurementReport

	Reason for change
	At Step 9  the RRCReconfiguration message releases the DelayValueConfig measurement, but the periodic measurements for reportQuantityCell (rsrp, rsrq..) is still active and therefore the UE shall send periodict measurement report with measresults with resultsSSB_Cell and no ul_PDCP_DelayValueResultList_r16


	Summary of change
	New function introduced to receive measurement report and assign a fail verdict when measurement report is received with DelayValue

	TTCN module
	NR_Measurement_Functions

	MCC160 Comment
	Accepted, but Prose CR needed



New Function:
	function f_NR_Check_NoDelayValueMeasurementReport(float p_Timer,
                                                      NR_CellId_Type p_NR_CellId,
                                                       template (present) MeasResults p_MeasResults,
                                                       template (present) MeasResults p_MeasResultsNoDelayValue
                                                      ) runs on NR_BASE_PTC
  {//@sic R5-201224 sic@ 
    timer t_Wait;
    
    t_Wait.start(p_Timer);
    alt {
      [] t_Wait.timeout
        {
        }
      [] SRB.receive(car_NR_SRB1_RrcPdu_IND(p_NR_CellId, cr_NR_MeasurementReport(p_MeasResults)))
        {
            t_Wait.stop;
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Check failed: Measurement Report received within a certain time");
        }
      [] SRB.receive( car_NR_SRB1_RrcPdu_IND( p_NR_CellId, cr_NR_MeasurementReport(p_MeasResultsNoDelayValue)))
        {
           repeat;
        }
    }
  };
  



2.6 [bookmark: _Toc110865494]f_NR_ReceivePeriodicMeasurementReport
	Function name
	f_NR_ReceivePeriodicMeasurementReport

	Reason for change
	When measurement configuration is configured with MeasConfig to setup ul-DelayValueConfig, the UE shall start to report a value for averageDelay_r16, therefore it is prposed to check at Step 5A that the PDCP-DelayValueResult is having a non zero value for the DRB where the data is loopedback (similarly for Step 8) 


	Summary of change
	Created a new function to test Step 5 and Step 8 measurement requirements and assign a pass verdict when the measurement report matches.

	TTCN module
	NR_Measurement_Functions

	MCC160 Comment
	Accepted, but Prose CR needed



New Function:
	// to receive measurement report with specific measurement results
    function f_NR_ReceivePeriodicMeasurementReport(NR_CellId_Type p_NR_CellId,
                                             template (present) MeasResults p_MeasResults
                                             ) runs on NR_BASE_PTC
    
    {
        alt
        {
            [] SRB.receive(car_NR_SRB1_RrcPdu_IND(p_NR_CellId, cr_NR_MeasurementReport(p_MeasResults)))
                {
                    f_NR_PreliminaryPass (__FILE__, __LINE__, "Received a measurement report");
                }
            [] SRB.receive( car_NR_SRB1_RrcPdu_IND( p_NR_CellId, cr_NR_MeasurementReport_Any ) )
              {
                repeat;
              }
        }
    }




1. [bookmark: _Toc54888065][bookmark: _Toc110865495][bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666]Branches executed
This NR5GC TC was executed in 5GS Stand Alone mode (Option 2) on NR Primary Band “n78” using NEA2 ciphering algorithm
4 [bookmark: _Toc110865496]Execution Log Files 
4.1 [bookmark: _Toc97714036][bookmark: _Toc110865497]MediaTek MT6983
The MediaTek MT6983 UE passed this test case on R&S® 5G Protocol Conformance Test platform. The documentation below is enclosed as evidence of the successful test case run [1]:
Test case execution log file:
TC_8_1_6_1_1_2_NR5GC_MTK.log
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)
In the log file the complete test case execution as well as the PICS/PIXIT parameter settings can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.
5 [bookmark: _Toc122434496][bookmark: _Toc295288973][bookmark: _Toc325725668][bookmark: _Toc110865498]References
	[1]
	R5s220962:   Supporting information for addition of NR5GC TC 8.1.6.1.1.2 .




